

Paper 177-31

## Dynamic Superscripts and Footnotes

Rex Tungala, Alcon Labs, Tx  
Rebecca Yost Montgomery, Alcon Labs, Tx

### ABSTRACT

This paper shows step by step how to create superscripts and matching footnotes from your data.

### INTRODUCTION

The initial challenge was to create footnotes with superscripts imbedded in the table for each parameter having missing data, but not every parameter contained missing data.

The TABULATE procedure will list the missing values. This will allow a data NULL datasets to create a FORMAT procedure for the superscripts and a TXT file for the actual footnotes. Another data step will be needed to connect the FORMAT value with the appropriate row. The REPORT procedure using a COMPUTE/CALL DEFINE statement will assign the FORMAT to a column from a NOPRINT column.

Create input data.

NOTE: Treatment1 will have 2 missing lab\_result  
 Treatment2 will have no missing lab\_result  
 Treatment3 will have 1 missing lab\_result

```
data create_dataa ;
    infile datalines missover;
    input treatment subjid lab_result;

    datalines;
1 101
1 102 15
1 103 32
1 104
2 105 18
2 106 31
2 107 31
3 108 19
3 109
3 110 24
;
run;
```

Input data

Obs	treatment	subjid	lab_ result
1	1	101	.
2	1	102	15
3	1	103	32
4	1	104	.
5	2	105	18
6	2	106	31
7	2	107	31
8	3	108	19
9	3	109	.
10	3	110	24

PROC TABULATE to generate mean and number of missing patients by treatment  
 output to SAS dataset  
 print to rtf text file

```
ods rtf body = "c:\temp\simple1.rtf" ; /* output to rtf */
title1 bold 'Simple Proc Tabulate Results';
proc tabulate data=create_dataa
              out=lab_result1(drop=_type_ _page_ _table_);
  class treatment;
  var lab_result;
  table treatment,
         lab_result*(mean*f=5. nmiss*f=5.);
run;
ods rtf close;
```

#### Simple Proc Tabulate Results

	lab_result	
	Mean	NMiss
treatment		
1	24	2
2	27	0
3	22	1

Generate a PROC FORMAT of treatment to PICTURE clause  
 Create a text file containing code and information for footnotes  
 Note: Watch the single quotes  
 Note: 97 is the ASCII equivalent of 'a'

```
ods escapechar='^';
filename line 'c:\temp\lines.txt';
filename super 'c:\temp\super.txt';
/* filename line catalog 'work.work.line.source'; */
/* filename super catalog 'work.work.super.source'; */

data null_;
  set lab_result1;

          /* a different format is created for
            each treatment - all data driven */

  file super;
  retain super 96;
  put 'proc format;';
  if lab_result_nmiss gt 0 then do;
    /* 97 is byte 'a' */
    super + 1;
    /* changes the numeric value of super to its byte value */
    superc=byte(super);
    put 'picture _fmt'
      treatment
          /* +(-1) will move print pointer one to the left */
      +(-1) ' _ '
      ;
    put @4 ' . = " ";
    put @4 'low - high = "0001^{super '
      superc
      +(-1) ' }";'
      ;
  end; else do;
    put 'value _fmt'
      treatment
      +(-1) ' _ '
      ;
    put @4 ' . = " ";
    put @4 'low - high = [8.]';
  end;
  put / 'run;';
```

```
/* write footnotes */
file line;
if lab_result_nmiss gt 0 then do;
  if lab_result_nmiss = 1 then put
    'line "^{super '
    superc
    +(-1) '}'
    lab_result_nmiss
    'patient had missing lab data.>";'
  ; else
  if lab_result_nmiss > 1 then put
    'line "^{super '
    superc
    +(-1) '}'
    lab_result_nmiss
    'patients had missing lab data.>";'
  ;
end;
run;

%include super; /* submit c:\temp\super.txt to reader */
```

```
c:\temp\super.txt
```

```
proc format;
picture _fmt1_
  . = " "
  low - high = "0001^{super a}";
run;

proc format;
value _fmt2_
  . = " "
  low - high = [8.];
run;

proc format;
picture _fmt3_
  . = " "
  low - high = "0001^{super b}";
run;
```

```
c:\temp\lines.txt
```

```
line "^{super a}2 patients had missing lab data.";
line "^{super b}1 patient had missing lab data.";
```

Assign superscript format to a SAS variable based on treatment

```
data lab_result2;
  set lab_result1;
  lab_result_mean = round(lab_result_mean);
  fmtname= "_fmt"||trim(left(put(treatment,8.)))||"_.";
run;
```

Data Ready for PROC REPORT

Obs	treatment	lab_result_ Mean	lab_result_ NMiss	fmtname
1	1	24	2	_fmt1_.
2	2	27	0	_fmt2_.
3	3	22	1	_fmt3_.

Format for Treatment name

```
proc format;
  value tmtname
    1 = 'Treatment 1'
    2 = 'Treatment 2'
    3 = 'Treatment 3'
  ;
run;
```

Proc Report using COMPUTE to assign the value of FMTNAME to the FORMAT for LAB\_RESULT.

A %INCLUDE will bring in the previously created footnotes.

```
ods rtf body = "c:\temp\desired1.rtf" ;
title1 'Desired Results';
proc report data=lab_result2 nowd;
    column treatment fmtname lab_result_mean ;

    define treatment / group 'Treatment' format=tmtname. center;
    define fmtname / group noprint ' ' ;
    define lab_result_mean / max 'Mean Lab Result' ;

    /* apply format from fmtname value to variable lab_result_mean */
    /* watch the quotes */
    compute lab_result_mean;
        call define('lab_result_mean','FORMAT',fmtname);
    endcomp;

    /* apply footnote */
    compute after / style={just=1};
        %include line; /* c:\temp\lines.txt */
    endcomp;
run;
ods rtf close;
```

#### Desired Result

Treatment	Mean Lab Result
Treatment 1	24 <sup>a</sup>
Treatment 2	27
Treatment 3	22 <sup>b</sup>
<sup>a</sup> 2 patients had missing lab data. <sup>b</sup> 1 patient had missing lab data.	



## CONCLUSION

Because SAS can generate a text file of code and then read that file back in and submit it:

PROC FORMAT can be created from input data  
FOOTNOTEs can be generated from input data  
Etc... macro calls by data values, error reports...

Because SAS PROC REPORT will allow CALL DEFINE statements to be values from datasets:

Formats can be data assigned by cell  
Etc...Significant values can be shown in different STYLES

## ABOUT THE AUTHORS

Rex Tungala and Rebecca Montgomery are SAS Programmers for Alcon Labs. They generate tables for FDA submissions.

They can be contacted at:

Rex Tungala [Rex.Tungala@AlconLabs.com](mailto:Rex.Tungala@AlconLabs.com)

Rebecca Yost Montgomery [Rebecca.Montgomery@AlconLabs.com](mailto:Rebecca.Montgomery@AlconLabs.com)

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.